

# スピログラフを再現しよう GIFANIM Device Driverを用いた アニメーション図形の作成



長谷川 要  
麒麟ビール株式会社 医薬カンパニー



# 要旨

- スピログラフとは、歯車のついた固定定規と可動定規からなりたっている。可動定規にはペンを差し込む穴があいており、ペンを可動円の穴に入れて回転させると様々な幾何学模様が描画される。これらスピログラフは数学的にはサイクロイドという名称で知られている。
- 本論文では、SAS/GRAPHのGIFANIM Device Driverを利用して、サイクロイド図形をコンピュータ上で動画により再現する方法について紹介する。



# スピログラフ


- 100円ショップで購入したスピログラフ描画用の定規



# スピログラフの原理

- スピログラフは、歯車のついた円形の穴の空いた固定定規の内側に、ペンをさした円形の可動定規を回転させることにより、図1に示すような幾何学模様を作成するための道具として昔から知られている。原理は簡単であるが、円形の可動定規の大きさや、ペンの差し込む位置を変えることにより、様々な異なる幾何学模様が作成できる。

# 図 1 . スピログラフの作図例






# 軌跡座標の数式解

- これらの図形の軌跡を数学的に表現するための一連の計算式は、サイクロイド（内トロコイド）という呼び名で定義された単純な計算式が知られている。図2に示すように、半径 $a$ の円Aに沿ってその内側を、半径 $b$ の円Bが滑ることなく回転する際に、半径 $b$ の円の中心から $c$ の距離にある点Pの描く軌跡の座標を $(x, y)$ とすると、各々の座標の値は式1によって与えられる。

## 図2.座標の模式図



## 式1.点Pの軌跡座標の計算式

$$x = (a - b)\cos(\theta) + c \cos\left(\frac{a - b}{b}\theta\right)$$

$$y = (a - b)\sin(\theta) - c \sin\left(\frac{a - b}{b}\theta\right)$$






# 軌跡の計算方法

- 式1を用い、 $a$ 、 $b$ 、 $c$ に定数を代入し、さらに回転角  $\theta$  の値を少しずつ変化させることにより、軌跡 $(x, y)$ の座標を順次計算することができる。現実の世界では、 $c$ の値は円Bの内側にしか存在できないため、 $c$ の取り得る値は $c < b$ と制約されるが、コンピュータ上では $c > b$ でも計算を行うことが可能となる。



# の設定方法

- 出発点から描画を開始し、元の出発点に戻るためには、条件により  $\theta$  に与える角度の範囲が異なります。円A、円Bの半径a、bの比を  $m=p/q$  としたときに、  $\theta$  に与える角度（ラジアン）は以下の通りとなります。
  - $m$ が整数の場合、 $0 < \theta < 2$
  - $m$ が有理数の場合、 $0 < \theta < 2q$
  - $m$ が無理数の場合、どんなに計算しても元の出発点に戻れません。



# SASによるスピログラフの作図


- SASを用いたスピログラフの作図にあたっては、完成した図形を静止画として表示させるのではなく、出来上がる過程を楽しむために動画アニメーションで表示することを試みた。
- SASのGraphics Device Driverにはアニメーション図形の作成に適したGIFANIM Driverが用意されており、これを用いることによりGIFアニメーションファイルが容易に作成できることからこの機能を利用することとした。



# GIFANIMの利用方法

- GIFANIM Device Driverの利用にあたっては、GOPTIONSステートメントで指定を行うことに利用が可能である。表1にGIFANIM Device Driverの利用方法についてその手順を示す。
- 静止画を作図する場合の手順とは異なりやや煩雑であるが、作図部分に用いるプロシジャなどは静止画を表示するものを変更せずにそのまま用いることができる。

# 表1. GIFANIMの利用手順





# GIFANIM Driverの詳細情報

- GIFANIM Device Driver: Animation Process. Web Technologies. SAS Institute Inc.  
<http://www.sas.com/rnd/web/internet/driver/GIFANIM/process.html>
- GIFANIM Device Driver: Controls. SAS Homepage. Web Technologies. SAS Institute Inc.  
<http://www.sas.com/rnd/web/internet/driver/GIFANIM/controls.html>



# 作図を行うプログラム

- 作図を行うプログラムは、描画に必要なパラメータ、すなわち円Aの半径a、円Bの半径b、ペンの位置c及び  $\theta$  の計算範囲とする角度t ( $0 < \theta < t$ ) を指定するだけで作図できるようにマクロプログラムを作成した。
- グラフ作図にあたっては、SAS/GRAPHソフトウェアが提供するGPLOTプロシジャのANNOTATEステートメントの機能を用いることとした。表2に今回作成したマクロのプログラムの引数を、また表3にプログラムコードを示す。
- プログラム中の%cycloidマクロのパラメータを変化させることにより様々なサイクロイド図形の作成が可能となる。



## 表2. %cycloidマクロ引数の説明

### ■ マクロの呼び出し

`%cycloid(a,b,c,t);`

### ■ マクロの引数

a: 外側の円の半径

b: 内側の円の半径

c: ペンの位置。

内側の円の中心からの距離


t: 計算範囲とする角度 ( 0                      t    )






## 表3. %cycloidマクロのソース

```
* ----- GIFANIMドライバーにより作成される.GIFファイルの出力先を設定;  
filename giffile 'd:¥gifanim.gif'; * 任意のファイル名を指定する。;  
* ----- 作図前処理に用いるマクロ;  
%macro initgif;    goptions reset=all;  
    goptions device=gifanim gsfname=giffile gsfmode=replace gcopies=0  
    gwait=25;  
%mend;  
* ----- 作図追加前処理に用いるマクロ;  
%macro outgif;  
    goptions gsfmode=append;  
%mend;  
* ----- 作図終了時の処理に用いるマクロ;  
%macro endgif;  
    goptions gepilog='3B'x;  
%mend;
```




```
* ----- サイクロイド図形作成用マクロ;  
%macro cycloid(a,b,c,t);  
  %let pi=3.1415926535;  
  %let pitch=100; * GIFアニメーションファイルのフレーム数;  
  %do loop=0 %to &pitch;  
    * ----- サイクロイドの座標演算;  
    data cycloid;  
      a=&a; b=&b; c=&c;  
      xsys='2'; ysys='2';  
      do th=0 to &t*&pi/&pitch*(&loop+1) by &pi/&pitch;  
        x=(a-b)*cos(th)+c*cos((a-b)*th/b);  
        y=(a-b)*sin(th)-c*sin((a-b)*th/b);  
        if th=0 then do;  
          function='MOVE';  
          output;  
        end;  
        else do;  
          function='DRAW';  
          output;  
        end;  
      end;  
    end;  
  run;
```



```
* ----- グラフ作成に用いる外枠の座標を含むダミーデータ;  
data flame;  
    x=-200; y=-200; output;  
    x=200; y=200; output;  
run;  
* ----- 出力設定;  
%if &loop=0 %then %initgif;  
%else %outgif;  
%if &loop=&pitch %then %endgif;  
* ----- 作図処理;  
proc gplot data=flame;  
    plot y*x/annotate=cycloid haxis=axis1 vaxis=axis1;  
    axis1 length=10cm; * 軸の長さを設定;  
    symbol1 v=none; * シンボルを無印にする;  
run;  
quit;  
%end;  
%mend;
```

`%cycloid(100,40,55,4);`



`%cycloid(250,175,50,14);`

